

Whats the Fuss?
Using Procedures
and Service
Programs



MAGiC

MID-ATLANTIC GROUP OF IBM i
COLLABORATORS

Jim Buck

Phone 262-705-2832

Email – jbuck@impowertechnologies.com

Twitter - @jbuck_imPower

www.impowertechnologies.com



imPOWER
TECHNOLOGIES

Powered by Jim Buck

How can imPower Technologies help your company?

IBM i Education

Online IBM i Classes: Unique offering

- IBM i Concepts
- Programming in ILE RPG

Onsite IBM i Classes:


- Topics include the latest RDi/RPG/SQL techniques
- Two-day hands-on lecture and exercises
- Optional Third day - Let's design and code a new application

Modernization: Getting started

- Helping a company getting started down the modernization road
- The thought process of modern development
- Learn to use these new tools and concepts



Rational Developer for i – 9.6.0.6



This is the FUTURE!

***** Disclaimer *****

All of the code that I show in my presentations is supported using RDi!

Modular Programming

Developing code in small, independent units offers several advantages

- Reusability
- Fewer errors
- Easier to test
- Changes are unlikely to cause unwanted side effects
- Easier team development

Callp (Call a Prototyped Procedure or Program) operation

- Passes control to separate *PGM object
- When called program returns, caller resumes

Prototyping the Call Interface

Variable's scope is limited to program in which it is defined

Call operation passes parameters to communicate values between caller and called program

- Parameters are values that one program can pass to another
- Called program can then accept parameters and use those values to control its processing

Prototype definition defines call interface

- Name of program to call
- Number of parameters, data attributes, order

Compiler uses prototype to call program correctly and to ensure caller passes correct parameters

Prototyping the Call Interface

Prototype definition has two parts

- Prototype header
- Parameter descriptions

Dcl-pr (Declare Prototype) instruction signals beginning of prototyped call interface

- Coded in Declarations section
- End-pr instruction ends prototype

```
Dcl-pr Updcust Extpgm('AR002');  
    Company Char(5);  
    Customer Zoned(7:0);  
End-pr;
```

Prototyping the Call Interface

Subsequent lines describe parameters

- Similar to data structure subfields
- May list up to 255 parameters

Parameters need not be named

- Variable must be declared elsewhere
- May use *N placeholder instead
- If name matches RPG reserved word, Dcl-
parm (Declare Parameter) instruction is
required
 - Optional otherwise

```
Dcl-pr Updcust Extpgm('AR002');  
    Company Char(5);  
    Customer Zoned(7:0);  
End-pr;
```

```
Dcl-pr Updcust Extpgm('AR002');  
    *N Char(5);  
    *N Zoned(7:0);  
End-pr;
```

```
Dcl-pr Updcust Extpgm('AR002');  
    Dcl-parm Company Char(5);  
    Dcl-parm Customer Zoned(7:0);  
End-pr;
```


Prototyping the Call Interface

If there are no parameters to pass, prototype may be coded on single line

```
Dcl-pr Sleep Extpgm End-pr;
```

Callp (Call Prototyped Procedure or Program)

Callp (Call Prototyped Procedure or Program) operation invokes associated *PGM object, then passes control to it

- Actually coding Callp is optional

Parameters in parentheses immediately follow prototype name

- If no parameters, code empty parentheses instead

```
Dcl-pr Updcust Extpgm('AR002');  
  *N Char(5);  
  *N Zoned(7:0);  
End-pr;  
...  
Callp Updcust(Company:Custnbr);
```

```
Updcust(Company:Custnbr);
```

```
Dcl-pr Sleep Extpgm End-pr;  
...  
Sleep();
```

Procedure Interface – Called Program

Called RPG program should also use prototype to describe list of parameters that it is to receive from caller

- Main procedure prototype

Prototypes in each program must match each other

- Same number of parameters, data attributes
- Can Should use /COPY to keep them in sync!
- Names need not be same

Called program must include procedure interface definition

- In addition to prototype
- Defines variable names to hold received parameter values

Procedure Interface

Dcl-pi (Declare Procedure Interface) instruction defines main procedure interface

- Similar to prototype definition
- Must appear after main procedure prototype

Two parts

- Procedure interface header
- Parameter definitions to hold received parameters

```
Dcl-pr AR002 Extpgm;  
  *N      Char(5);  
  *N      Zoned(7:0);  
End-pr;
```

```
Dcl-pi AR002;  
  Company Char(5);  
  Customer Zoned(7:0);  
End-pi;
```

Passing Parameters by Reference

Default parameter passing method

RPG passes parameter by passing address of memory location represented by variable

- Called program uses that storage address to retrieve and process the parameter value

Both programs share same storage

- If called program changes parameter value,
- Caller recognizes change when it regains control

Passing Parameters by Reference

```
// caller
Dcl-pr Nextpgm Extpgm;
  *N Ind;
End-pr;

Dcl-s Okay Ind Inz(*Off);
...
           // okay = *off before call
Nextpgm(Okay);
           // okay = *on after call
...

```

```
// Nextpgm
Dcl-pr Nextpgm Extpgm;
  *N Ind;
End-pr;

Dcl-pi Nextpgm;
  Flag *Ind;
End-pi;

...
Flag = *On;           // Change value of Flag
*Inlr = *On;
Return;

```

Passing Parameters by Read-only Reference

Alternative method of passing parameters

- **Const keyword** on prototype (and interface) specifies read-only reference

Several advantages over passing by reference

- System offers protection against called program changing parameter values
- Parameter values need not be represented by variable
- Parameter data types need not precisely match prototype

```
Dcl-pr Addcust Extpgm('AR001');  
  *N Char(9)      Const;  
  *N Zoned(7:0)  Const;  
End-pr;  
...  
Addcust('BBY' : Lastcustomer + 1);  
...
```

Passing Parameters by Read-only Reference

Callp can pass variables, literals, or expressions

Caller can first evaluate expression or literal, and then make temporary copy of value before invoking called program

Caller passes storage address of temporary copy

```
Dcl-pr Addcust Extpgm('AR001');  
  *N Char(9)      Const;  
  *N Zoned(7:0)  Const;  
End-pr;  
...  
Addcust('BBY' : Lastcustomer + 1);  
...
```


Passing Parameters by Read-only Reference

```
Dcl-pr Addcust Extpgm('AR001');  
  *N Char(9)      Const;  
  *N Zoned(7:0)  Const;  
End-pr;  
...  
Addcust('BBY' : Lastcustomer + 1);  
...
```

```
// Called program (AR001)  
...  
Dcl-pr Addcust Extpgm('AR001');  
  *N Char(9)      Const;  
  *N Zoned(7:0)  Const;  
End-pr;  
  
Dcl-pi Addcust;  
  Company Char(5)      Const;  
  Customer Zoned(7:0) Const;  
End-pi;  
...  
  // (Some processing goes here...  
  // cannot change Company or Customer)  
...  
*Inlr = *On;  
Return;
```

Choosing the Parameter Passing Method

If caller needs to access changes made by called program, pass by reference

- Or if large number of parameters must be passed
 - Improves the program “Call” performance

Use read-only reference as preferred method for passing parameters between programs

- Improves coding flexibility (variables, literals, or expressions)
- Protects integrity of caller’s data

Fitting the Pieces

Caller includes two items

- *Prototype definition* to describe parameters to pass
- Call operation to execute call

Called program contains three items

- Main *procedure prototype definition* to describe parameters to receive
- Main *procedure interface* to accept parameters and place them into variables
- Return operation to return control to caller
 - Generally, will first set *Inlr *On

Dynamic Program Calls and Static Binding

*PGM to *PGM calls are dynamic calls

- Association between caller and called program is resolved at runtime
- Programs are never physically connected

Integrated Language Environment enables static calls

- Association between modules is resolved before runtime
 - When *PGM is created
- No need for resolution at runtime
- Procedures are key component of static binding

Introduction to Procedures

A **Procedure** is self-contained, identifiable collection of RPG statements within *PGM object

- Performs specific task and returns to caller
- Not a system object

Modern RPG Program is comprised **four sections**

- **Control options**
 - Provides default options for program
- **Declarations**
 - Identifies and defines files, variables, and other data items
- **Main procedure**
 - Processes, calculations, and procedures
- **Subprocedures**
 - Declarations and processes for optional distinct program functions that main procedure or other subprocedures can execute once or many times

Role of Procedures in a Program

Like a subroutine, but with more capabilities

Subprocedure can be created independently from program

- Code in separate source member and compile into separate module, then bind it when you create program

Subprocedure enables flexible variable scoping

- Global declarations make all variables equally accessible by entire program
 - Only option without subprocedures
- Local variables are recognized only within procedure in which they are defined
 - Data values are communicated between procedures by passing parameters

Subprocedure supports return value

- User-defined function

Subprocedures can be recursive

- Can call itself

Coding a Procedure

Dcl-proc (Declare Procedure) instruction begins procedure

- End-proc instruction ends it

Subprocedure code can include

- Procedure interface
- Declarations
- Processing statements (calculations)

```
// -----  
// Procedure Celsius =  
//           Converts Fahrenheit to Celsius  
// -----  
  
Dcl-proc Celsius;  
  
// ----- Procedure interface  
Dcl-pi *N Zoned(5:2);  
       Fahrenheit Zoned(5:2);  
End-pi;  
  
// ----- Local variables  
Dcl-s Temperature Zoned(5:2);  
  
Eval(H) Temperature = (5/9)  
                    * (Fahrenheit - 32);  
Return Temperature;  
  
End-proc Celsius;
```

Coding a Procedure

Need not name subprocedure interface

- Use *N placeholder instead

Dcl-proc instruction includes data attributes of return value

- Can be value of any supported data type
- Can also be data structure (using Likeds) or array (using Dim)

```
// -----
// Procedure Celsius =
//           Converts Fahrenheit to Celsius
// -----

Dcl-proc Celsius;

// ----- Procedure interface
Dcl-pi *N Zoned(5:2);
       Fahrenheit Zoned(5:2);
End-pi;

// ----- Local variables
Dcl-s Temperature Zoned(5:2);

Eval(H) Temperature = (5/9)
                       * (Fahrenheit - 32);
Return Temperature;

End-proc Celsius;
```


Coding a Procedure

Declarations within subprocedure have local scope

- Including variables in procedure interface
- Restricted to subprocedure
- No other procedure will recognize local variables

Return operation includes value to return to caller

- May be variable, literal, or expression

```
// -----  
// Procedure Celsius =  
//           Converts Fahrenheit to Celsius  
// -----  
  
Dcl-proc Celsius;  
  
// ----- Procedure interface  
Dcl-pi *N Zoned(5:2);  
       Fahrenheit Zoned(5:2);  
End-pi;  
  
// ----- Local variables  
Dcl-s Temperature Zoned(5:2);  
  
Eval(H) Temperature = (5/9)  
                      * (Fahrenheit - 32);  
Return Temperature;  
  
End-proc Celsius;
```

Coding a Procedure

Subprocedure prototype is coded in global section of source

Prototype names procedure

- If procedure name is different from prototype name, use **Extproc** keyword

Prototype includes return value

```
// ----- Global Area of Pgm
Dcl-pr Celsius Zoned(5:2);
      *N Zoned(5:2);
End-pr;
// -----
// Procedure Celsius =
//       Converts Fahrenheit to Celsius
// -----

Dcl-proc Celsius;

// ----- Procedure interface
Dcl-pi *N Zoned(5:2);
      Fahrenheit Zoned(5:2);
End-pi;

// ----- Local variables
Dcl-s Temperature Zoned(5:2);

Eval(H) Temperature = (5/9)
                        * (Fahrenheit - 32);
Return Temperature;

End-proc Celsius;
```

Coding a Procedure

Locally scoped variables are allocated in automatic storage

- Reinitialized each time procedure executes
- Do not retain value between iterations

Static keyword uses static storage instead


- Retains value

```
Decl-s Counter Uns(5) Static;
```

RDi and creating PR and PI's

RDi has some great features to help you get started with Prototypes and Interfaces.

Most of the problems are caused by a “Piece(s)” missing.



Creating PR and PI's

RPG Procedure Wizard

RPG Procedure
Create RPG procedure

Procedure type: Subprocedure

Procedure name: MyProcedure

External name (EXTPROC):

Purpose: This is a new procedure

Exportable for use with other code (EXPORT)

Run only one thread in this procedure at any one time (SERIALIZE)

Generate program-interface information (PCML)

Do not generate program-interface information (PCML)

Parameters created for this procedure:

| Name | Type | Length | Decimal | Keywords | Comments |
|-----------|----------|--------|---------|----------|----------|
| StartDate | D - Date | | | | |
| EndDate | D - Date | | | | |

Buttons: Add... Duplicate... Change... Remove Move Up Move Down

Pass the operational descriptors with the parameters

Return a value

Refresh Outline view upon completion

Generate free-form D-spec and P-spec code

Generate free-form C-spec

Navigation: Back Next > Finish Cancel



Add Procedure Parameter

Parameter name: InvoiceCost

Parameter description: Cost of the invoice

Defined like:

Field Data Structure Record Format

Type: S - Zoned Decimal

Length: 7 1 - 63

Decimals: 2 0 - 7

The class of the object:

CCSID: 0 - 65535

Fractional seconds: 0 - 12

(MATH):

Separator:

Sample:

Array entries (DIM):

Other keywords: ALTSEQ(*NONE) ASCEND DESCEND NOOPT

Parameter passing options: Pass copy of contents (VALUE) Pass read-only reference (CONST)

Parameter options (OPTIONS):

Optional (*NOPASS) Optional (*OMIT)

Variable size (*VARSIZE) Null-terminated string (*STRING)

Right adjusted string (*RIGHTADJ) Trim before passing (*TRIM) (>=V5R3)

Pass null indicator (*NULLIND) (>=V5R4)

Buttons: OK Cancel

Executing a Procedure

Callp operation calls procedure without return value

- Or if program ignores return value

Function-like call uses return value

```
callp Updcust(Company:Custnbr);
```

```
Metrictemp = Celsius(Englishtemp);
```

```
If Celsius(Englishtemp) > 100;  
...  
Endif;
```

Cycle Main Programs

Most traditional RPG programs are cycle main programs

Cycle main program has main procedure implicitly specified

- Main procedure is not named
 - Location in the program code designates main procedure
- Includes main source section (the main procedure), followed by zero or more subprocedure sections
 - Main procedure is everything before **first Dcl-proc** instruction
 - May include local subprocedures coded following main procedure

Compiler automatically includes RPG cycle to provide program initialization, termination, file input and output (I/O)

Sample Cycle Main Program

```
// Control Options =====*  
Ctl-Opt Option(*NoDebugIO) DftActGrp(*No);  
  
Dcl-pr CallTemp ExtPgm;  
  *N char(5);  
End-pr;  
  
Dcl-pi CallTemp;  
  CharTemp char(5);  
End-pi;  
  
// ----- Prototypes  
Dcl-pr Celsius Zoned(5:2);  
  *N Zoned(5:2);  
End-pr;
```



```
// ----- Global variables
Decl-s Message      Char(52);
Decl-s Metrictemp   Zoned(5:2);
Decl-s State        Varchar(8);
Decl-s EnglishTemp  Zoned(5:2);
// ----- Main procedure

EnglishTemp = %Dec(CharTemp:5:2);
Metrictemp = Celsius(Englishtemp);

Select;
  when Metrictemp < 0;
    State = 'solid';
  when Metrictemp = 0;
    State = 'freezing';
  when Metrictemp = 100;
    State = 'boiling';
  when Metrictemp > 100;
    State = 'gaseous';
  other;
    State = 'liquid';
Endsl;
```

```
Message = 'At ' + %Char(Englishtemp) + ' degrees (' +
          %Char(Metrictemp) + ' Celsius), water is ' +
          State + '.';
Dsply Message '*REQUESTER';
*Inlr = *On;
Return;

// -----
// Procedure Celsius Converts Fahrenheit to Celsius
// -----
Dcl-proc Celsius;

// ----- Procedure interface
Dcl-pi *N Zoned(5:2);
      Fahrenheit Zoned(5:2);
End-pi;

// ----- Local variables
Dcl-s Temperature Zoned(5:2);

      Eval(H) Temperature = (5/9) * (Fahrenheit - 32);
      Return Temperature;

End-proc Celsius;
```

Linear Main Programs

Linear main program explicitly names main procedure

- Ctl-opt instruction uses Main keyword to name main procedure

Global section in linear main program does not contain any executable code

- Before first **Dcl-proc**
- Can include global declarations

Compiler will not embed RPG cycle into linear main program

Program implicitly initializes variables, locks data areas, and opens files when program starts

Program does not use *Inlr to trigger automatic shutdown

- Resources are not cleaned up or closed when program ends, unless program explicitly does so

Program is recursive

Sample Linear Main Program

```
Ctl-opt Option(*NoDebugIO) DftActGrp(*No);
Ctl-opt Main(Driver);

// Control Options =====*
Dcl-pr Driver ExtPgm('CALLTEMP1');
    *N char(5);
End-pr;

// ----- Prototypes
Dcl-pr Celsius Zoned(5:2);
    *N Zoned(5:2);
End-pr;
// -----
// Main procedure
//
Dcl-proc Driver;
----- Procedure interface
Dcl-pi *N;
    CharTemp char(5);
End-pi;
```

Sample Linear Main Program

```
// ----- Local variables
Decl-s Message      Char(52);
Decl-s Metrictemp   Zoned(5:2);
Decl-s State        Varchar(8);
Decl-s EnglishTemp  Zoned(5:2);
// ----- Main procedure

EnglishTemp = %Dec(charTemp:5:2);
Metrictemp = Celsius(Englishtemp);

Select;
When Metrictemp < 0;
  State = 'solid';
When Metrictemp = 0;
  State = 'freezing';
When Metrictemp = 100;
  State = 'boiling';
When Metrictemp > 100;
  State = 'gaseous';
Other;
  State = 'liquid';
Endsl;
```

Sample Linear Main Program

```
Message = 'At ' + %Char(Englishtemp) + ' degrees (' +
          %Char(Metrictemp) + ' Celsius), water is ' +
          State + '.';
Dsply Message '*REQUESTER';

Return;
End-proc Driver;

// -----
// Procedure Celsius = Converts Fahrenheit to Celsius
// -----
Dcl-proc Celsius;
// ----- Procedure interface
Dcl-pi *N Zoned(5:2);
    Fahrenheit Zoned(5:2);
End-pi;

// ----- Local variables
Dcl-s Temperature Zoned(5:2);
Eval(H) Temperature = (5/9) * (Fahrenheit - 32);
Return Temperature;

End-proc Celsius;
```

Nomain Modules

Nomain modules consists of program segments without no main procedure

Source consists only of global declarations and subprocedures

Can combine with other modules to create program

One module in program must have main procedure

Includes Ctl-opt instruction with Nomain keyword

Enhance code reusability

Can help enforce business rules and practices by centralizing application functions

Eliminate redundant code

Improve maintainability, reliability

Nomain Modules

Dcl-proc “Export” keyword allows procedure to be called from outside module

- Other modules in program can execute procedure, even though they don’t contain the code for the procedure
- Without Export, procedure can only be executed from within module
- Most procedures in Nomain modules include Export keyword



CELSIUS - Nomain Module


```
Ctl-opt Nomain;  
// Copy Block - Replaces including the Prototype  
  
    /Copy RPGTRAIN/QRPGLESRC,CALLTMPCPY  
  
Dcl-proc Celsius Export;  
  
// ----- Procedure interface  
Dcl-pi *N Zoned(5:2);  
    Fahrenheit Zoned(5:2);  
End-pi;  
  
// ----- Local variables  
Dcl-s Temperature Zoned(5:2);  
  
Eval(H) Temperature = (5/9) * (Fahrenheit - 32);  
Return Temperature;  
  
End-proc Celsius;
```



Creating Modular Programs

CRTRPGMOD (Create RPG Module) command compiles source member, creates *MODULE object

- Module contains compiled, executable code
- Cannot run module by itself
- Module is interim building block for eventual program object



Using /COPY, /INCLUDE

/Copy and /Include functions tell compiler to include source records from another source member (Copybook)

- Can reuse single copy of code without having to retype
- Useful for storing prototypes and other reusable code snippets

```
/Copy Mylib/Mysource,Prototypes
```

Copybook member

Copybook source file

```
/Include Mylib/Mysource,Prototypes
```

Using /COPY, /INCLUDE

Both directives use the same syntax and have the same purpose

- /copy is expanded by the SQL precompiler
- /include is ignored by the SQL precompiler

```
/Copy Mylib/Mysource,Prototypes
```

Copybook member

Copybook source file

```
/Include Mylib/Mysource,Prototypes
```

CALLTEMP2 - Linear Main Program

```
// Control Options =====*
      Ctl-opt Main(Driver) Option(*NoDebugIO);

// Copy Block
      /Copy RPGTRAIN/QRPGLESRC,CALLTMPCPY
// -----
// Main procedure
// -----
Dcl-proc Driver;
// ----- Procedure interface
Dcl-pi *N;
      CharTemp char(5);
End-pi;
// ----- Global variables
      Dcl-s Message      Char(52);
      Dcl-s Metrictemp   Zoned(5:2);
      Dcl-s State        Varchar(8);
      Dcl-s EnglishTemp  Zoned(5:2);
```

(Continued...)

```
// ----- Main procedure
EnglishTemp = %Dec(CharTemp:5:2);
Metrictemp = Celsius(Englishtemp);

Select;
  when Metrictemp < 0;
    State = 'solid';
  when Metrictemp = 0;
    State = 'freezing';
  when Metrictemp = 100;
    State = 'boiling';
  when Metrictemp > 100;
    State = 'gaseous';
  Other;
    State = 'liquid';
Endsl;


Message = 'At ' + %Char(Englishtemp) + ' degrees (' +
          %Char(Metrictemp) + ' Celsius), water is ' +
          State + '.';
Dsply Message '*REQUESTER';
Return;
End-proc Driver;
```

(Continued...)

Creating CALLTEMP2 - Module

CRTRPGMOD (Create RPG Module) command compiles source member, creates *MODULE object

- Module contains compiled, executable code
- Cannot run module by itself
- Module is interim building block for eventual program object



The screenshot shows the 'Create RPG Module (CRTRPGMOD)' dialog box. The fields are filled with the following values:

- Module: CALLTEMP2
- Library: RPGTRAIN
- Source file: QRPGLSRC
- Library: RPGTRAIN
- Source member: CALLTEMP2
- Source stream file: (empty)
- Generation severity level: 10
- Text description: *SRCMBRTXT
- Compiler options: *EVENTF
- Debugging views: *SOURCE
- Replace module: *YES

At the bottom, the command line is displayed: CRTRPGMOD MODULE(RPGTRAIN/CALLTEMP2) SRCFILE(RPGTRAIN/QRPGLSRC) SRCMBR(CALLTEMP2) OPTION (*EVENTF) DBGVIEW(*SOURCE) REPLACE(*YES)


Creating Modular Programs

CRTRPGMOD (Create RPG Module) command compiles source member, creates *MODULE object

- Module contains compiled, executable code
- Cannot run module by itself
- Module is interim building block for eventual program object

CRTPGM (Create Program) command binds module to *PGM object


- Bind-by-copy
- Program is runnable, using CALL command




Creating Modular Programs

CRTPGM command can combine one or more modules during binding

- Modules may have been written using any ILE language



Creating the Executable Program




The screenshot shows the 'Create Program (CRTPGM)' dialog box. The 'Program' field is set to 'CALLTEMP2' and the 'Library' is 'RPGTRAIN'. The 'Module' field is empty. The 'Program entry procedure module' is set to 'CALLTEMP2' in the 'RPGTRAIN' library. The 'Text description' is '*ENTMODTXT'. A blue arrow points to the 'Advanced Parameters' section, which contains the text '*ENTMODTXT'. The text 'Default *FIRST' is written on the arrow. The dialog has 'OK', 'Restore defaults', and 'Cancel' buttons at the bottom.

Maintaining Modular Programs

Program modification requires compile and binding steps

UPDPGM (Update Program) command performs abbreviated binding step

- Lists only module(s) to replace in original program
- Unchanged modules are unaffected



Introduction to Service Programs

Service program (*SRVPGM) is code toolbox that many programs can use

- Binder need not physically copy subprocedure code into each client program
- Bind-by-reference

Service program does not have main procedure

- Any subprocedure can be entry point into service program
- Multiple entry point program



Introduction to Service Programs

Single entry point ILE program—or another service program—can invoke any exported procedure in service program

- Only one copy of actual subprocedure code exists
- Many other programs (clients) share

Combine some performance advantages of bind-by-copy static binding with modularity and flexibility benefits of dynamic program call



Compiling and Binding Service Programs

Service program procedures have no unique coding requirements

- Use Nomain modules
- Source can have global declarations section
 - Items declared in global section, before any subprocedures, are available to all procedures within module
 - Includes prototype for each procedure

Service program procedures usually use “Export” keyword to ensure availability to client programs

- May hide procedure inside module by omitting “Export” keyword

Compile source code with CRTRPGMOD command


- Creates *Module object

Compiling and Binding Service Programs

We are using the “Celsius” Module that we created previously to create this service program.

CRTSRVPGM (Create Service Program) command binds multiple entry point service program

The actual CL command is shown in the window at right!



Additional Service Prgms

Advanced Parameters

```
CRTSRVPGM SRVPGM(RPGTRAIN/CELSIUS) MODULE(RPGTRAIN/CELSIUS) EXPORT(*ALL)
```

Compiling and Binding Service Programs

MODULE parameter lists modules to copy into *Srvpgm object

- No entry module

EXPORT allows you to reference a source member with a list of procedures to export. I have selected *ALL

BNDSRVPGM parameter lists other service programs to bind by reference to this service program

- If procedures refer to other procedures in other service programs

BNDDIR parameter supports binding directories

- Allows binder to find necessary modules or service programs not explicitly listed with the MODULE or BNDSRVPGM parameters

End result of successful binding step is *Srvpgm object

Deploying Service Programs in an Application

Single entry point program – or another service program – can invoke any service program procedure

- Caller does not call service program itself
- Caller calls procedure instead

```
Callp Updcust(Company:Custnbr);
```

```
Metrictemp = Celsius(Englishtemp);
```

```
If Celsius(Englishtemp) > 100;  
...  
Endif;
```

Deploying Service Programs in an Application

CRTPGM command decides whether to use bind-by-copy or bind-by-reference (service program)

- CRTSRVPGM command also uses same logic

Modules listed in MODULE parameter are bound by copy

Service programs listed in BNDSRVPGM parameter are bound by reference

| | | | |
|--------|----------------------|---|---------------------|
| CRTPGM | PGM(THISPGM) | + | |
| | MODULE(THISPGM) | + | ← Bind-by-copy |
| | ENTMOD(THISPGM) | + | |
| | BNDSRVPGM(DATSRVPGM) | + | ← Bind-by-reference |
| | BNDDIR(MYBNDDIR) | | |

Using RDi to create a Program


PGM - program name will be "CELSIUSPGM"

MODULE - "CALLTEMP2" Driver module, we created in a previous step

BNDSVRPGM - "MYSRVPGM" the service program we created previously

ACTGRP - *CALLER it will run in the same Activation group as the program that called "CELSIUSPGM"

We now have a functional (runnable) program



| | | | | |
|---------------------------------|-----------|---|--------------------|-----------------|
| Program: | PGM | > | CALLTEMP3 | Name |
| Library: | | > | RPGTRAIN | Name |
| Module: | MODULE | > | | Name, generic* |
| Library: | | > | | Name |
| | | | RPGTRAIN/CALLTEMP3 | |
| Text 'description': | TEXT | | *ENTMODTXT | Character value |
| Program entry procedure module: | ENTMOD | | *FIRST | Name |
| Library: | | | *LIBL | Name |
| Bind service program: | BNDSVRPGM | | | Name, generic* |
| Service program: | | | CELSIUS | Name |
| Library: | | | RPGTRAIN | |
| Activation: | | | *IMMED | |
| Binding directory: | BNDDIR | | *NONE | Name |
| Library: | | | *LIBL | Name |
| Activation group: | ACTGRP | | *ENTMOD | Name |
| Creation options: | OPTION | | | |
| Listing detail: | DETAIL | | *NONE | |

Advanced
 All Parameters
 Keywords


CRTPGM PGM(RPGTRAIN/CALLTEMP3) MODULE(RPGTRAIN/CALLTEMP3) BNDSVRPGM((RPGTRAIN/CELSIUS))

Maintaining Service Programs

Modification to service program procedure requires compile-then-bind process to apply changes

UPDSRVPGM (Update Service Program) command abbreviates binding step

- List only changed module(s)



Update Service Program (UPDSRVPGM)

Service program: > CELSIUS Name
 Library: > RPGTRAIN Name
 Module: > Name, generic*
 Library: > Name

RPGTRAIN/CELSIUS Add Remove Move up Move down

Export: *CURRENT Name
 Export source file: QSRVSRC Name
 Library: *LIBL Name
 Export source member: *SRVPGM Name

Export source stream file: Path name

Advanced All Parameters Keywords

OK Restore defaults Cancel

Service Program Signatures

Signature is service program attribute identifying its external interface

- Usually system-generated string

When service program is bound to client, client makes note of service program signature

When client loads at runtime, system matches signatures

- Mismatch causes program load to fail

Service program has only one current signature, but can retain many previous signatures

- Binder uses current signature
- If clients are using previous signature, they can continue without recompiling or rebinding

Service Program Signatures

DSPSRVPGM command shows all valid signatures

- **DETAIL(*SIGNATURE)**

```
Display Service Program Information                                Display 9 of 10
Service program . . . . . : WWWRPG
Library . . . . . :
Owner . . . . . :
Service program attribute . . . . . : RPGLE
Detail . . . . . : *SIGNATURE

Signatures:

00000C3F1C2150BA678CBE3335CE8A46
00000000C3F1C21A31ED2A61A3CE8A46
00000000000C3F100155AF3567988A46

F3=Exit  F11=Display character signature  F12=Cancel  F17=Top  F18=Bottom
```

Service Program Signatures

```
Display Program Information
Display 4 of 7
Program . . . . . : WWZTEST   Library . . . . . :
Owner . . . . . :
Program attribute . . : RPGLE
Detail . . . . . : *SRVPGM

Type options, press Enter.
 5=Display

Service
Opt Program   Library   Activation Signature
-  WWWRPG     *IMMED  000000C3F1C2DD823B7AB476E5CE8A4C
-  QRNXIE     QSYS    *IMMED  D8D9D5E7C9C54040404040404040404
-  QRNXIO     QSYS    *IMMED  D8D9D5E7C9D64040404040404040404
-  QRNXDUMP   QSYS    *IMMED  D8D9D5E7C4E4D4D7404040404040404
-  QLEAWI     QSYS    *IMMED  4470FABA08585397BDF0CF195F82EC

F3=Exit  F4=Prompt  F11=Display character signature  F12=Cancel  F17=
F18=Bottom
```

```
Display Service Program Information
Display 9 of 10
Service program . . . . . : WWWRPG
Library . . . . . :
Owner . . . . . :
Service program attribute . . . . . : RPGLE
Detail . . . . . : *SIGNATURE

Signatures:

00000C3F1C2150BA678CBE3335CE8A46
00000000C3F1C21A31ED2A61A3CE8A46
00000000000C3F100155AF3567988A46

F3=Exit  F11=Display character signature  F12=Cancel  F17=Top  Bottom
F18=Bottom
```

Must match

Using Binder Language

Binder language describes service program's signature

- Stored in source file

Three commands

- STRPGMEXP (Start Program Export List) begins signature
- EXPORT (Export a Program Symbol) names procedure(s)
- ENDPGMEXP (End Program Export List) ends signature

```
STRPGMEXP PGMLVL(*CURRENT)
EXPORT SYMBOL(WWWDECODE)
EXPORT SYMBOL(WWWDUMP)
EXPORT SYMBOL(WWWECHO)
EXPORT SYMBOL(WWWEXTRACT)
ENDPGMEXP
```


Using Binder Language

Refer to binder language source when creating service program

- CRTSRVPGM ... EXPORT(*SRCFILE)
SRCFILE(filename)
SRCMBR(member)

```
CRTSRVPGM SRVPGM(WWRPG) +  
          MODULE(WWDECODE WWWDUMP WWWECHO WWWEXTRACT) +  
          EXPORT(*SRCFILE) +  
          SRCFILE(QSRVSRC) +  
          SRCMBR(WWRPG)
```

Using Binder Language

```
STRPGMEXP PGMLVL(*CURRENT)
  EXPORT SYMBOL(WWWDECODE)
  EXPORT SYMBOL(WWWDUMP)
  EXPORT SYMBOL(WWWECHO)
  EXPORT SYMBOL(WWWGETDOC)
  EXPORT SYMBOL(WWWREAD)
  EXPORT SYMBOL(WWWREPLACE)
ENDPGMEXP
```

```
STRPGMEXP PGMLVL(*PRV)
  EXPORT SYMBOL(WWWDECODE)
  EXPORT SYMBOL(WWWDUMP)
  EXPORT SYMBOL(WWWECHO)
  EXPORT SYMBOL(WWWGETDOC)
ENDPGMEXP
```

```
STRPGMEXP PGMLVL(*PRV)
  EXPORT SYMBOL(WWWDECODE)
  EXPORT SYMBOL(WWWDUMP)
  EXPORT SYMBOL(WWWECHO)
ENDPGMEXP
```

Binder language manages service program's current signature and any previous signatures

DO NOT REMOVE OR REARRANGE
entries in existing signatures

- Add new procedures to end of signature block
- Current signature block must contain same procedures in same order as previous signature(s)

Using Binder Language

May explicitly name signature

- Up to 16 characters

Service program needs only one named signature block

- Add new procedures to end of named signature
- Existing programs will still run without rebinding

```
STRPGMEXP PGMLVL(*CURRENT) SIGNATURE(WWRPG)
EXPORT SYMBOL(WWWDECODE)
EXPORT SYMBOL(WWDUMP)
EXPORT SYMBOL(WWECHO)
EXPORT SYMBOL(WWWGETDOC)
EXPORT SYMBOL(WWWREAD)
EXPORT SYMBOL(WWWREPLACE)
ENDPGMEXP
```

Exporting Data Items

Export keyword can declare variable or data structure whose contents are available across modules

- Only one module actually allocates storage
 - Export keyword
- Other modules use data item allocated by exporting module
 - Import keyword

May be useful to export data items from service program and then import them into client

```
// Module: PRIMARY  
Dcl-s Mydata Packed(9:2) Export Inz(200);  
...  
Proca();
```

```
// Module: PROCA  
Dcl-s Mydata Packed(9:2) Import;  
...
```

Subfile Application Example

```

A - 5250 Display
File Edit View Communication Actions Window Help
Host: IDEVUSR030.IDEV.CLOUD.COM Port: 23 Workstation ID: Disconnect
Program ID: PROG172SQL CloudServices24x7, Inc. 5/16/18
Customer Name Generic Inquiry Screen

Opt. A=Add, C=Change, D=Delete
Last name First Name Street City ST ZipCode
-----
Albright Scotty 8040 STATE ST. CHICAGO IL 60635-1209
Alvaordo Dennis 447 W. DARTMOO GURNEE IL 60031-3136
ABDUL HALIM NARIZA 5652 N. 46TH S KALAMAZOO MI 49008-0000
AMERINE MICHAEL 789 S. ASH LAWRENCE TX 76550-0000
Brenneman Jim 111 32ND AVE. BOYCE LA 71409-0000
Buckley Jimmy 1234 Great Lak Kenosha WI 53140-1346
BARRY TRACY 32348 S. 39TH GRAND RAPIDS MI 49501-0002
BAYONNE ALFREDO 10423 S.E. 30T BELLEVUE WA 98007-0012
CHO DEUK HWAN 1234 WEST ST. LOCKHART SD 29364-0000
COIN DOREEN 302 WASHINGTON WHITE PLAINS NY 71530-0039
COOK MICHAEL 2478 E. MAIN S ABILENE TX 79604-1110
Dylan Bobby 124 RAIN STREE CAMBRIDGE MA 21421-3450
DAVIS JEFF 23 5TH STREET KENSOHA WI 51231-1234
DEWAARD MICHAEL 4532 SYCAMORE PHILADELPHIA PA 19104-0121

More...

53 rows fetched from cursor CUSTNAMECUR.
F3=Exit F09=Reload Subfile F12=Cancel

MA A 06/005
IDEVUSR030.IDEV.CLOUD.COM:23
  
```

Subfile Application – Putting it Together

Comprised
of three
programs +
copybook

- **PROG172SQL** – Main Driver Program
 - Runs the 5250 screens
 - Handles the CREATE, READ, UPDATE and DELETE Logic
- **CUSTSRVPGM** – Service program that handles SQL I/O
 - SQL INSERT, UPDATE, SELECT and DELETE Code
 - Returns data Structures (Customer and SQL Status)
- **GETSQLDIAG** – Service program that:
 - Processes the GET DIAGNOSTICS command
 - Puts the results into a data structure
 - Returns this data structure to the calling program



PROG172SQL – Driver program

```

CUSTSRVPGM.SQLRPGLE  GETSQLDIAG.SQLRPGLE  PROG172SQL.SQLRPGLE
Line 43      Column 86      Replace
.....1.....2.....3.....4.....5.....6.....7.....8.....9.....
001500      //   Written: Jim Buck  Email: j buck@impowertechnologies      *
001600      //   Copyright 2017 - imPower Technologies                    *
001700      //   These examples are for demonstration purposes. There are *
001800      //   NO Express of implied warranties. Have fun with the code! *
001900      //   *****
002000      //   =====
002100      //   file spec
002200      //   =====
002300      //   *** CUSTOMER Table just used for AssignNextNbr ***
002400      Dcl-f CUSTOMER DISK(*ext) keyed Usage(*Update:*Delete:*Output);
002500      Dcl-f DisplayScreen WORKSTN Qualified Alias ExtFile(*extdesc) UsrOpn
002600           ExtDesc('PROG172D') sfile(custsfl:rrn) indds(indicators);
002700
002800      // Copy in the Prototypes
002900      /copy RPGTRAIN/QSRVPGMSRC,CUSTSRVPGC
003000
003100      EXEC SQL
003200          Set Option
003300              Naming = *Sys,
003400              Commit = *None,
003500              UsrPrf = *User,
003600              DynUsrPrf = *User,
003700              Datfmt = *iso,
003800              CloSqlCsr = *EndMOD;
003900
004000      // =====
004100      // definition specs
004200      // =====
004300      Dcl-Ds indicators ;
004400          Exit          Ind Pos(03);          // F3=Exit
004500      i ReloadSFL      Ind Pos(09);          // F9=Reload Subfile
004600          FkeyPressed  Ind Pos(10) Inz(*off);
004700          Cancel       Ind Pos(12);


```

Outline

type filter text

- Control Statements
- Global Definitions
 - Files
 - Data Structures
 - Fields
 - Constants
 - Indicators
 - Prototypes
- Main Procedure
- Subprocedures
 - LoadSFL
 - DisplaySFL
 - ProcessSFL
 - ClearSFL
 - HandleSQLMessages
 - AddRecord
 - ChangeRecord
 - DeleteRecord
 - ClearFields

CUSTSRVPGM – SQL Database I/O



The screenshot shows an IDE with three tabs: CUSTSRVPGM.SQLRPGLE, GETSQLDIAG.SQLRPGLE, and PROG172SQL.SQLRPGLE. The main editor displays the source code for CUSTSRVPGM.SQLRPGLE. The code includes comments, control statements, and declarations for data structures and arrays. The Outline window on the right lists the following items:

- Control Statements
- Global Definitions
 - Data Structures
 - Fields
 - Constants
 - Indicators
 - Prototypes
- Subprocedures
 - GetCUSTOMER_Data : EXPORT
 - GetCUSTOMER_DataRecds : EXPORT
 - GetCUSTOMERRecs_DynSelect : EXPOF
 - BuildSQLStmt
 - DeleteCUSTOMER_Data : EXPORT
 - UpdateCUSTOMER_Data : EXPORT
 - WriteCUSTOMER_Data : EXPORT
 - GetNewCustNbr : ZONED(6: 0)

GETSQLDIAG – SQL Database I/O

```

CUSTSRVPGM.SQRLPGLE  GETSQLDIAG.SQRLPGLE  PROG172SQL.SQRLPGLE
Line 19      Column 1      Replace
.....1.....2.....3.....4.....5.....6.....7.....8.....9.....0
000128      // Used to process SQL GET DIAGNOSTICS Service program *
000129      // Written: Jim Buck  Email: jrbuck@impowertechnologies *
000130      // Copyright 2017 - imPower Technologies *
000131      // These examples are for demonstration purposes. There are *
000132      // NO Express of implied warranties. Have fun with the code! *
000133      // *****
000134      // *****
000135      // Service program GETSQLDIAG
000136      /Include RPGTRAIN/QSRVPGMSRC,GETSQLDIGC
000140
000141      // Data Structure for SQL Results
000142      Dcl-Ds UtilDSSQL inz;
000143      MessageId Char(10);
000144      MessageId1 Char(7);
000145      MessageId2 Char(7);
000146      MessageLength int(5);
000147      MessageText Char(120);
000148      ReturnedSQLCode int(5);
000149      ReturnedSQLState Char(5);
000150      RowsCount int(10);
000151      SuccessFlag Ind; // Operation was Successful
000152      End-Ds;
000153
000154      Dcl-proc GetDiagnostics Export;
000155      Dcl-Pi *N;
000156      DiagUtilDS LikeDS(UtilDSSQL);
000157      End-Pi ;
000158
000159      Clear DiagUtilDS;
000160
000161      Exec sql GET DIAGNOSTICS CONDITION 1
000162      :DiagUtilDS.MessageId      = DB2_MESSAGE_ID,
000163      :DiagUtilDS.MessageId1    = DB2_MESSAGE_ID1,


```

Outline

type filter text

- Control Statements
- Global Definitions
 - Data Structures
 - Fields
 - Indicators
 - Prototypes
- Subprocedures
 - GetDiagnostics : EXPORT

CUSTSRVCPY – Prototype Copybook



The screenshot shows a code editor window with the following tabs: CUSTSRVPGM.SQLRPGLE, GETSQLDIAG.SQLRPGLE, PROG172SQL.SQLRPGLE, and CUSTSRVPGC.RPGLE. The main editor displays a copybook with the following content:

```


Line 1      Column 1      Replace
.....1.....2.....3.....4.....5.....6.....7.....8.....9.....0
000101      // *****
000103      // Prototypes used for CUSTOMER CRUD Application      *
000104      // Written: Jim Buck  Email: jbuck@impowertechnologies  *
000105      // Copyright 2017 - imPower Technologies      *
000106      // These examples are for demonstration purposes. There are  *
000107      // NO Express of implied warranties. Have fun with the code!  *
000108      // *****
000121
i 000135      DCL-PR GetCUSTOMER_Data;
000137      CUSTOMERDataDS LIKEDS(CUSTOMER_IODataDS);
000138      WrkCustNbr Zoned(6:0);
000139      WrkUtilDS LikeDS(UtilDSSQL);
000140      END-PR ;
000141
i 000142      DCL-PR GetCUSTOMER_DataRecds;
000143      CUSTOMER_IORcDsDS_LENGTH int(10);
000145      CUSTOMER_IORcDsDS LIKEDS(CUSTOMER_IODataDS) Dim(9999);
000146      WrkUtilDS LikeDS(UtilDSSQL);
000147      END-PR ;
000148
i 000149      DCL-PR GetCUSTOMERRecs_DynSelect;
000150      CUSTOMERDataDS_LENGTH int(10);
000151      CUSTOMERDataDS LIKEDS(CUSTOMER_IODataDS) Dim(9999);
000153      SearchDS LIKEDS(SearchRecDS);
000154      WrkUtilDS LikeDS(UtilDSSQL);
000155      END-PR ;
000156
i 000157      Dcl-Pr BuildSQLStmt;
000158      Statement varChar(4096);
000159      SearchDS LIKEDS(SearchRecDS);
000160      END-PR;
000161
i 000162      DCL-PR DeleteCUSTOMER_Data;

```

The Outline window on the right shows the following structure:

- Global Definitions
- Prototypes
 - GetCUSTOMER_Data : EXTPROC ('GETCUS
 - GetCUSTOMER_DataRecds : EXTPROC ('G
 - GetCUSTOMERRecs_DynSelect : EXTPROC
 - BuildSQLStmt : EXTPROC ('BUILDSQLSTM
 - DeleteCUSTOMER_Data : EXTPROC ('DELE
 - UpDateCUSTOMER_Data : EXTPROC ('UPE
 - WriteCUSTOMER_Data : EXTPROC ('WRITI
 - GetNewCustNbr : ZONED(6:0) EXTPROC (

Questions or Comments?



Jim Buck
Phone 262-705-2832
Email – jbuck@impowertechnologies.com
Twitter - @jbuck_imPower
www.impowertechnologies.com

MAGiC
MID-ATLANTIC GROUP OF IBM i
— COLLABORATORS —